Arduino Lesson ( )

Use of soil moisture sensor

Name: _____ ( ) Class: _____ Date: _____

Objectives: At the end of this lesson, you would be able to
1. Connect a Arduino Uno microcontroller to receive readings from a soil moisture sensor, using Analog and Digital output modes.
2. Connect an ESP32 to obtain readings from the soil moisture sensor.
3. Write sketches for ESP32 to send email alert via Blynk app.
4. Write sketches for ESP32 to send email alert via Thingspeak platform.

Apparatus:
- 01 Arduino UNO microcontroller & USB cable
- 01 soil moisture sensor (FC-28 or YL-69/HL-69)*
- jumper wires



Image source: https://shopee.sg/Soil-Humidity-Hygrometer-Moisture-Detection-Sensor-Module-For-Arduino-i.50203676.1279852868

Note:
- * Refer **Annex A**
- The **soil moisture sensor** measures the volumetric content of water inside the soil and gives us the moisture level as output. It works based on the **resistance** of the soil between the 2 probes which is affected by the soil moisture level.
- How the soil moisture sensor works? https://lastminuteengineers.com/soil-moisture-sensor-arduino-tutorial/ *The **more water** in the soil, the **lower its resistance**, the **lower the output** voltage reading from the sensor.*
- **Specifications**:

Input voltage	Output voltage	Input current	Output signal
3.3 – 5 V	0 – 4.2 V	35 mA	Both analog & digital

- The sensor consists of a probe and a module. It is equipped with both analog and digital outputs, so it can be used in both **analog and digital mode**.



Image source: https://randomnerdtutorials.com/guide-for-soil-moisture-sensor-yl-69-or-hl-69-with-the-arduino/

Assignment 1: Set up soil moisture sensor with Arduino Uno in **analog output mode**

1. Plug the Arduino Uno into the USB port on the laptop installed with Arduino software.
2. Select **Tools → Board**: scroll down to select "**Arduino Uno**"
3. and select **Port**: "COM#"
4. In the analog mode, connect the sensor to the microcontroller:

Soil moisture sensor	→	**Arduino Uno microcontroller**
VCC	→	5 V
A0	→	A0
GND	→	GND (ground)

5. The analog output from the sensor gives a value from 0-1023. The moisture is measured as a percentage. Map these values to 0 -100% and show these on the serial monitor.

Note

- When the sensor is switched on, the power LED on the module will light up.
- Do a calibration by observing the serial monitor readings and place the probes in air and water respectively. Example: probes in air: 294 == moisture 0%; probes in container of water: 180 == moisture 100%

Reference:

- http://www.circuitstoday.com/arduino-soil-moisture-sensor



Diagram drawn in tinkercad.com (circuits) (soil moisture sensor added separately)
Soil moisture sensor image adapted from https://lastminuteengineers.com/wp-content/uploads/arduino/Soil-Moisture-Sensor-Pinout.png

Assignment 1: Sample sketch (Analog output mode)

```
// soil moisture sensor in Analog Mode
int sensor_pin = A0;      // sensor input at analog pin A0
int output_value;
void setup() {
  Serial.begin(9600);     // initialize the serial monitor
  Serial.println("Reading from the sensor ...");
  delay(2000);
  }
void loop() {
  output_value= analogRead(sensor_pin);
  // map sensor output (soil resistance) to moisture percentage
  // dry soil resistance 294 -> moisture 0%, wet soil 180 -> 100%
  // suggested values are to be changed according to calibration
  output_value = map(output_value, 294, 180, 0, 100);

  Serial.print("Moisture : ");
  Serial.print(output_value);
  Serial.println("%");
  delay(1000);
  }
```



Assignment 1 serial monitor readings during calibration

Assignment 2: Calibrate & control the soil moisture sensor in **analog output mode**

6. In the analog mode, connect the sensor to the microcontroller:

Soil moisture sensor	→	**Arduino Uno microcontroller**
VCC	→	D7 (instead of 5 V)
A0	→	A0
GND	→	GND (ground)

7. Changes compared to Assignment 1 include:

- Use D7 to switch on/off the soil moisture sensor before taking a reading
- Calibrate the sensor in 3 ranges based on whether soil is “wet” or “dry”

Reference:

- http://www.circuitstoday.com/arduino-soil-moisture-sensor
- https://randomnerdtutorials.com/guide-for-soil-moisture-sensor-yl-69-or-hl-69-with-the-arduino/



Diagram drawn in tinkercad.com (circuits) (soil moisture sensor added separately)
Image adapted from https://lastminuteengineers.com/wp-content/uploads/arduino/Soil-Moisture-Sensor-Pinout.png

Assignment 2: Sample sketch (Analog output mode)

```
// soil moisture sensor in Analog Mode
// to switch the sensor ON/OFF using Uno and calibrate it

int sensor_pin = A0;  // sensor input at analog pin A0, OR #define sensor_pin A0
int sensor_power = 7;  // pin D7 will switch ON or OFF the sensor

/* Change these values based on your calibration values */
int soilWet = 150;   // Define max value we consider soil 'wet'
int soilDry = 220;   // Define min value we consider soil 'dry'

void setup() {
  pinMode(sensor_power, OUTPUT);
  digitalWrite(sensor_power, LOW);  // initially keep the sensor OFF

  Serial.begin(9600);
  Serial.println("Reading from the sensor ...");
  delay(2000);
  }

void loop() {

//get the reading from the function below and print it
  int moisture = readSensor();

  Serial.print("Soil moisture (Analog output): ");
  Serial.println(moisture);

  // map sensor output (soil resistance) to moisture percentage
  // calibrate dry soil resistance 320 -> moisture 0%, wet soil 30 -> 100%
  // values to be changed according to calibration
  int moisture_percentage = map(moisture, 320, 30, 0, 100);

  Serial.print("Soil moisture: ");
  Serial.print(moisture_percentage);
  Serial.println("%");

  // Determine moisture status of the soil
  if (moisture < soilWet) {
    Serial.println("Status: Soil is too wet");
  } else if (moisture >= soilWet && moisture < soilDry) {
    Serial.println("Status: Soil moisture is perfect");
  } else {
    Serial.println("Status: Soil is too dry - time to water!");
  }

  delay(1000);    // delay for 1 second
  Serial.println();
  }

// This function returns the analog soil moisture reading
int readSensor() {
  digitalWrite(sensor_power, HIGH);  // Turn the sensor ON
  delay(10);              // Allow power to settle
  int output_value = analogRead(sensor_pin);  // Read the analog value from sensor
  digitalWrite(sensor_power, LOW);   // Turn the sensor OFF
  return output_value;    // Return analog moisture reading
}
```

Note

- A common issue with a soil moisture sensor is its short lifespan when exposed to a moist environment. Applying power to the probe constantly (VCC = 5 V) increases the rate of corrosion. To overcome this, power it only when taking readings, so use the Arduino Uno to turn on/off the sensor as needed.
- With some trial and error when running the sketch, do a calibration by observing the serial monitor readings and obtain some threshold readings to trigger required actions for a particular type of soil.
  Example: < 150 soil is too wet; 150-220 is the target range; > 220 soil is dry enough to be watered



Assignment 2 set-up



Assignment 2 serial monitor readings during calibration

Note: Refer to **Annex B** on setting up soil moisture sensor with Arduino Uno in **digital output mode**

Annex A

Choice of type of soil moisture sensors

- Video: Why most Arduino soil moisture sensors suck? https://youtu.be/udmJyncDvw0
  - Resistive probe type of sensor undergoes corrosion (electrolysis) and damage easily when immersed in moisture and gives unreliable readings.
  - Capacitive type of sensor is more reliable and does not need to be in contact with water (can be placed in a plastic bag).

Annex B

Setting up soil moisture sensor with Arduino Uno in digital output mode

- *The moisture sensor module contains a potentiometer used to set a **threshold value**. The LM393 comparator on the module compares this threshold value with the sensor output value and then gives an **output** through the **digital pin** D0.*
- *The potentiometer allows for sensitivity adjustment of the digital output (DO). Rotate the knob clockwise to increase sensitivity and counterclockwise to decrease it.*
- *If the sensor output value > threshold value, the digital pin will give 5V and the LED (on D13) will light up.*
- ***Example** of application: when the moisture level in the soil crosses a threshold, activate a relay to start pumping water.*

Reference:

- https://lastminuteengineers.com/soil-moisture-sensor-arduino-tutorial/

***Note**: The digital output mode of use may not have the required sensitivity/reliability as the analog mode.*